Math 277
Ordinary Differential Equations
Fall 2005

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 9

Applications of First Order Equations

Initializations

```
> restart;
```

9.1 Applications of First Order Equations

Applications of first order differential equations appear in a variety of disciplines. Even though it is not possible to provide a uniform road map for the creation of mathematical models in such diverse environments, a few overall principles apply.

1) Read the problem very carefully.
2) When helpful, use pencil and paper to make a picture.
3) Clearly identify your independent and dependent variables and parameters.
4) Clearly identify a positive direction and an origin for your dependent variable.
5) Use well established principles in the applicable discipline to create the differential equation model.
6) Be aware that the parameters as well as the differential equation may differ according to which part of the domain of the independent variable is under consideration.
7) Identify your differential equation as separable, linear, exact, or neither. Then decide on a solution strategy.
8) Whenever possible find an exact solution of the problem. Only when such solution cannot be achieved should you turn to a numeric differential equation solver.
9) Use appropriate technology to execute the necessary calculations.
10) A plot of the results may be very illustrative.

Example 9.1.1

A 50 kg shell is fired straight upward from a gun with muzzle speed of 2000 m/sec. Let v = v(t) denote the vertical velocity of the shell at time t. The magnitude of the airresistance equals 0.05 v^2. The accelleration due to gravity is assumed to be g = 9.81 m/sec^2.

a) Compute the velocity function v(t) of the shell, valid between the moment of firing and the moment that maximum height is reached.
b) When does the shell reach maximum height?
c) What is the maximum height reached by the shell?
d) Compute the velocity function v(t) of the shell, valid between the moment that maximum height is reached, and the moment that the shell reaches the ground.
e) When does the shell hit the ground?
f) With what velocity does the shell hit the ground?
g) Plot the velocity of the shell from the moment of firing to the moment of impact with the ground.
h) Compute the limiting velocity of the shell on the way down. At impact, what percentage of its limiting velocity has been reached by the shell?
i) Plot the height of the shell from the moment of firing to the moment of impact with the ground.

Part a.

Let v(t) denote the velocity of the shell at time t. We measure v(t) vertically with positive direction upward. Let m denote the mass of the shell, while b denotes the proportionality constant of the air resistance term. The upward part of the shell's journey is ruled by the initial value problem

$$m\left(\frac{d}{dt}\mathrm{v}(t)\right) = -m\,g - b\,\mathrm{v}(t)^2$$
$$\mathrm{v}(0) = 2000$$

Observe that the differential equation is separable and can be solved accordingly.

```
> dequp:=m*diff(v(t), t)=-m*g-b*v(t)^2;
  pars:={m=50, g=9.81, b=0.05};
  deq_up1:=subs(pars, dequp);
```

$$dequp := m\left(\frac{d}{dt}\mathrm{v}(t)\right) = -m\,g - b\,\mathrm{v}(t)^2$$
$$pars := \{m = 50, g = 9.81, b = 0.05\}$$
$$deq\_up1 := 50\left(\frac{d}{dt}\mathrm{v}(t)\right) = -490.50 - 0.05\,\mathrm{v}(t)^2$$

Separate the variables and integrate.

```
> deq_up2:=deq_up1/rhs(deq_up1);
```

$$deq\_up2 := \frac{50\left(\frac{d}{dt}\mathrm{v}(t)\right)}{-490.50 - 0.05\,\mathrm{v}(t)^2} = 1$$

```
> sol_up1:=map(int, deq_up2, t)+(0=c);
```

$$sol\_up1 := -10.09637555\,\arctan(0.01009637555\,\mathrm{v}(t)) = t + c$$

Implement the initial condition and compute the integration constant c.

```
> eq_c:=subs({t=0, v(t)=2000}, sol_up1);
  val_c:=isolate(eq_c, c);
```

$$eq\_c := -10.09637555\,\arctan(20.19275110) = c$$
$$val\_c := c = -15.35975778$$

Update the solution and code the velocity of the shell as a Maple function v_up.

```
> sol_up2:=isolate(subs(val_c, sol_up1), v(t));
```

$$sol\_up2 := \mathrm{v}(t) = -99.04544409\,\tan(0.09904544409\,t - 1.521314030)$$

```
> v_up:=unapply(rhs(sol_up2), t);
```

$$v\_up := t \rightarrow -99.04544409\,\tan(0.09904544409\,t - 1.521314030)$$

```
>
```

Part b

The shell reaches maximum height when its velocity equals zero. Make a picture to see where approximately the zero occurs. Then use a numeric equation solver to accurately compute the time when maximum height is obtained.

```
> plot(v_up(t), t=0..20, v=-10..50);
```




```
> eqmax_t:=v_up(t)=0;
  tmax:=fsolve(eqmax_t, t=12..16);
```

$$eqmax\_t := -99.04544409 \tan(0.09904544409\, t - 1.521314030) = 0$$

$$tmax := 15.35975778$$

Maximum height is reached after 15.35975778 seconds.

Part c

Between the moment of firing and the reaching of maximum altitude, the height x(t) of the shell at time t, is given by

$$x(t) = \int_0^t v(u)\, du = \int_0^t v\_up(u)\, du$$

Here x(t) is measured with the positive direction upward and the origin at ground level.

```
> x_up:=unapply(int(v_up(u), u=0..t), t);
Warning, unable to determine if
152131403000/9904544409+100000000000/9904544409*Pi*_Z2+50000000000/9904544409*Pi is between 0 and
t; try to use assumptions or set _EnvAllSolutions to true
```

$$x\_up := t \to \int_0^t -99.04544409 \tan(0.09904544409\, u - 1.521314030)\, du$$

Maple produces a warning to alert us that we should not integrate over one of the singularities of the tangent function and therefore leaves the integral unevaluated. We can put Maple's worries to rest by adding the option "continuous", to the integration command. We update the definition of x_up.

```
> x_up:=unapply(int(v_up(u), u=0..t, continuous), t);
```

$$x\_up := t \to 3006.548430 - 500.\ \ln(1. + \tan(0.09904544409\, t - 1.521314030)^2)$$

```
> xmax:=x_up(tmax);
```

$$xmax := 3006.548430$$

The maximum height reached by the shell is 3006.548430 meters.

Part d

When the shell starts falling down, the differential equation changes, because the velocity changes sign and the air resistance always needs to point in the opposite direction of the velocity. The new initial value problem becomes.

$$m\left(\frac{d}{dt} v(t)\right) = -m\, g + b\, v(t)^2$$

$$v(t_{max}) = 0$$

```
> deqdwn:=m*diff(v(t), t)=-m*g+b*v(t)^2;
  pars;
  deq_dwn1:=subs(pars, deqdwn);
```

$$deqdwn := m\left(\frac{d}{dt} v(t)\right) = -m\, g + b\, v(t)^2$$

$$\{m = 50, g = 9.81, b = 0.05\}$$

$$deq\_dwn1 := 50\left(\frac{d}{dt} v(t)\right) = -490.50 + 0.05\, v(t)^2$$

Once again, the differential equation is separable and can be solved in the usual way.

Separate the variables and integrate.

```
> deq_dwn2:=deq_dwn1/rhs(deq_dwn1);
```

$$deq\_dwn2 := \frac{50\left(\frac{d}{dt} v(t)\right)}{-490.50 + 0.05\, v(t)^2} = 1$$

```
> sol_dwn1:=map(int, deq_dwn2, t)+(0=c);
```

$$sol\_dwn1 := -10.09637555 \operatorname{arctanh}(0.01009637555\, v(t)) = t + c$$

Implement the initial condition and compute the integration constant c.

```
> eq_c:=subs({t=tmax, v(t)=0}, sol_dwn1);
  val_c:=isolate(eq_c, c);
```

$$eq\_c := -10.09637555 \operatorname{arctanh}(0.) = 15.35975778 + c$$

$$val\_c := c = -15.35975778$$

Update the solution and code the velocity of the shell as a Maple function v_dwn.

```
> sol_dwn2:=isolate(subs(val_c, sol_dwn1), v(t));
```

$$sol\_dwn2 := v(t) = -99.04544409 \tanh(0.09904544409\, t - 1.521314030)$$

```
> v_dwn:=unapply(rhs(sol_dwn2), t);
```

$$v\_dwn := t \rightarrow -99.04544409 \tanh(0.09904544409\, t - 1.521314030)$$

```
>
```

Part e

Of course the shell hits the ground when its height equals zero. When the shell is falling that height is given by

$$x(t) = x_{max} + \int_{t_{max}}^{t} v(u)\, du = x_{max} + \int_{t_{max}}^{t} v\_dwn(u)\, du$$

Again x(t) is measured with the positive direction upward and the origin at ground level.

```
> x_dwn:=unapply(xmax+int(v_dwn(u), u=tmax..t), t);
```

$$x\_dwn := t \rightarrow 3006.548430 - 1570.796327\, I + 500.\ \ln(\tanh(0.09904544409\, t - 1.521314030) - 1.) + 500.\ \ln(\tanh(0.09904544409\, t - 1.521314030) + 1.)$$

Observe there are complex parts in the expression above, partly due to numerical round off. There are a variety of remedies for this phenomena. One of the simplest is to build Maple's Re function into the expression for x_dwn. Re (x) computes the real part of x. We update the definition of x_dwn.

```
> x_dwn:=unapply(Re(xmax+int(v_dwn(u), u=tmax..t)), t);
```

$$x\_dwn := t \rightarrow 3006.548430 + 500.\ \ln(|\tanh(0.09904544409\, t - 1.521314030) - 1.|) + 500.\ \ln(|\tanh(0.09904544409\, t - 1.521314030) + 1.|)$$

First make a sketch, then use a numeric equation solver to find an accurate value for the time of impact.

```
> plot(x_dwn(t), t=tmax..60, labels=[t,x]);
```




```
> eqimpact_t:=x_dwn(t)=0;
  timpact:=fsolve(eqimpact_t, t=50..60);
```

$$eqimpact\_t := 3006.548430 + 500.\ \ln(|\tanh(0.09904544409\, t - 1.521314030) - 1.|) + 500.\ \ln(|\tanh(0.09904544409\, t - 1.521314030) + 1.|) = 0$$

$$timpact := 52.70709321$$

The time of impact equals 52.70709321 seconds.

Part f

```
> vimpact:=v_dwn(timpact);
```

$$vimpact := -98.92421256$$

The shell hits the ground with a velocity of −98.92421256 meters per second. Of course that corresponds to a speed of 98.92421256 meters per second.

Part g

Use Maple's piecewise routine to combine the way up and the way down.

```
> V:=unapply(piecewise(0<=t and t<=tmax, v_up(t), t>tmax and t<=timpact, v_dwn(t),
  undefined), t):
  V(t);
```

$$\begin{cases} -99.04544409 \tan(0.09904544409\, t - 1.521314030) & 0 \le t \text{ and } t \le 15.35975778 \\ -99.04544409 \tanh(0.09904544409\, t - 1.521314030) & 15.35975778 < t \text{ and } t \le 52.70709321 \\ \textit{undefined} & \textit{otherwise} \end{cases}$$

```
> plot(V(t), t=0..timpact, axes=frame, view=[-5..timpact+5, -200..2010],
  labels=[t,v]);
```




```
>
```

Part h

The limiting velocity of the shell is given by

$$\lim_{t \to \infty} v(t) = \lim_{t \to \infty} v\_dwn(t)$$

```
> vlimit:=limit(v_dwn(t), t=infinity);
```

$$vlimit := -99.04544409$$

```
> %_reached:=vimpact/vlimit*100;
```

$$\%\_reached := 99.87760009$$

At impact, the shell has reached 99.87760009 % of its limiting velocity.

```
>
```

Part g

Similar to what we did with the velocity functions, we combine the position functions using the piecewise command. In this case however, the formulas are too long to completely and nicely display on the screen. To save space, we create a special representation of XD(t) with fewer decimal places for display purposes only.

```
> X:=unapply(piecewise(0<=t and t<=tmax, x_up(t), t>tmax and t<=timpact, x_dwn(t),
  undefined), t):
  XD:=unapply(piecewise(0<=t and t<=evalf(tmax, 3), evalf(x_up(t), 2),
  t>evalf(tmax, 3) and t<=evalf(timpact, 3), evalf(x_dwn(t), 2), undefined), t):
  XD(t);
```

$$\begin{cases} 3000. - 500. \ln(1. + \tan(0.099\, t - 1.5)^2) & 0 \le t \text{ and } t \le 15.4 \\ 3000. + 500. \ln(|\tanh(0.099\, t - 1.5) - 1.|) + 500. \ln(|\tanh(0.099\, t - 1.5) + 1.|) & 15.4 < t \text{ and } t \le 52.7 \\ \textit{undefined} & \textit{otherwise} \end{cases}$$

```
> plot(X(t), t=0..timpact, axes=frame, view=[-5..timpact+5, -200..3010],
  labels=[t,x]);
```


3000
2500
2000
1500
1000
500
0
x
0
10
20
30
40
50
t

>